
APPLICATIONS ON THE INTERRUPT SYSTEM 

 

 

1. THE BOARD AND THE APPLICATION 

 

The application bellow is an example of how the interrupt system can be used. This 

example shows the major aspects, like the interrupt redirect, its solving and the development of a 

handler routine. 

  On IRQ7 input (the less prioritary input, destined for LPT, parallel port) of the 8259A 

controller (master) there are a three possibilities to generate an interrupt: with a switch, from 

channel 2 of a I8253 circuit with address 100h (see fig. 1) and from a telephone disc (see fig.2) 

The interrupt source selection is done with a jumper (JP2). 

Fig. 1 Electric scheme of the circuit 

 

For the interrupt source 1 the application INT7.ASM presented bellow was developed. The 

application counts the interrupts arrived on line ‘1’ from the pushbutton (and cleaned with a D 

bistable ) and signals it on the screen. 

 
;------------------------ 

PAGE 60,132 

TITLE INT7 

COMMENT * Generates interrupts from a pushbutton on IRQ7* 

STIVA SEGMENT  PARA STACK 'STACK' 

DW  512 DUP(?) 

STIVA ENDS 

;--------------------------------------------- 

DATA SEGMENT  PARA PUBLIC 'DATA' 

irq7_old   dw  2 DUP (?)    ; Old irq7 interrupt address 

mes_in     db 0dh,0ah,"Push the key connected to IRQ7 ",0dh,0ah, "$" 

mes_out    db 0dh,0ah,"Done",0dh,0ah, "$" 

caracter   db 30h 

DATA ENDS 

;----------------- 

CODE SEGMENT   PARA PUBLIC 'CODE'  

MAIN PROC FAR 

ASSUME CS:CODE, DS:DATA, SS:STIVA ,ES:NOTHING 

push ds 

xor ax,ax 

push ax        ;for return 

mov ax,data 

mov ds,ax 

mov ah,9            ; message display, string display function   

mov  dx,offset mes_in     ;message address offset 

int  21h                  ; DOS interrupt call 

cli 

;-- Redirect intrerrupt IRQ7 to user’s routine -------- 

mov ax,350fh        ;Read timer interrupt address in ES:BX 

int 21h              ; DOS interrupt call 

mov irq7_old,bx    ;saveold interrupt offset address 

mov irq7_old+2,es     ;and segment addres 

push ds  

mov dx,offset irq7int   ;new routine offset address  

mov ax,seg irq7int      

mov ds,ax  

mov ax,250fh    ;load address from DS:DX in TVI  

int 21h               ; DOS interrupt call 

pop ds 

mov ax,0600h    ;screen empty 

  mov cx,0 

  mov dx,164fh 

  mov bh,7 

  int 10h 

mov al,00111000b    ;masks settings 

out 21h,al 

sti 

stai:  mov ah,1   ;wait for a key press 

     int 16h 

     jz  stai 

cli      ;--Old interrupt irq7---   

push ds               ;saves  DS 

mov ax,250fh          ;restores vector in TVI 

lds dx,dword ptr irq7_old  ;loads old address in   DS:DX 

int 21h                    ; DOS interrupt call 

pop ds                ;restores DS 

mov al,10111000b   ;restores mask  

out 21h,al   

sti 

mov ah,9    ;-- end message --------- 

mov dx,offset mes_out 

int 21h 

mov ah,02 

mov dl,07        ;sound at exit 

int 21h  

mov ax,4C00h         ;end program through  

int 21h              ; a DOS function call 

MAIN ENDP 

 

IRQ7INT   PROC  NEAR ;-- new interrupt routine--- 

push ax 

push ds 

mov ax,data 

mov ds,ax 

mov ah,0eh 

mov al,caracter 

int 10h   ;character display 

inc caracter 

mov al,7   ; generates sound  

int 10h  

mov al,20h 

out 20h,al 

pop ds 

pop ax 

iret    ;return 

 

IRQ7INT    ENDP 

;== End ===================================== 

code ends        ; COD segment end 

end main        ; program end 

;*-------------------------------------------------------*; 

 

 

 2. WORK 
 

a. Analyze INT7.ASM application and run it. 

b. Rewrite the application: 

- using the controller in polling mode 

- using software cleaning instead of the bistable 

c. Adapt the application for the second interrupt source (channel 2 - timer) 

d. The third interrupt source is a telephone disc. Write an application that takes the impulses 

from it and displays the dialed number. Attention to cleaning! 

 Fig. 2 The electric scheme of the telephone disc connection to IRQ7 


